Low Power Design Using PIC16/17

Author: *Rodger Richey*
Logic Products Division

INTRODUCTION

Power consumption is an important element in designing a system, particularly in today's battery powered world. The PIC16/17 family of devices has been designed to give the user a low-cost, low-power, and high performance solution to this problem. For the application to operate at the lowest possible power, the designer must ensure that the PIC16/17 devices are properly configured. This application note describes some design techniques to lower current consumption, some battery design considerations, and suggestions to assist the designer in resolving power consumption problems.

DESIGN TECHNIQUES

Many techniques are used to reduce power consumption in the PIC16/17 devices. The most commonly used methods are SLEEP Mode or external events. These modes are the best way to reduce Ipd in a system. The PIC16/17 device can periodically wake-up from Sleep using the Watchdog Timer or external interrupt, execute code and then go back into SLEEP Mode. In SLEEP Mode the oscillator is shut off, which causes the PIC16/17 device to consume very little current. Typical Ipd current in most PIC16/17 devices is on the order of a few microamps.

In cases where the PIC16/17 uses an RC oscillator but cannot use SLEEP Mode, another technique is used to lower power consumption. An I/O pin can remove a parallel resistance from the oscillator resistor while waiting for an event to occur. This would slow down the internal clock frequency, by increasing the resistance, and thus reduce Ipd. Once an event occurs the resistor can be switched in and the PIC16/17 device can process the event at full speed. Figure 1 shows how to implement this technique. The resistor R1 would be used to increase the clock frequency by making the I/O pin an output and setting it to VDD.

FIGURE 1: USING AN EXTERNAL RESISTOR TO LOWER POWER IN RC MODE



External events can be used to control the power to PIC16/17 devices. For these cases, the Watchdog Timer can be disabled to further reduce current consumption. Figure 2 shows an example circuit that uses an external event to latch power on for the PIC16/17 device. Once the device has finished executing code, it disables power by resetting the latch. The latching circuit uses a low-power 4000 series CMOS quad chip which consumes a typical of 10 µA of current. The measured value of current consumption for the complete circuit with the PIC16/17 powered-down was 1 nA. Current consumption for a PIC16/17 in SLEEP Mode is typically 1 µA.

FIGURE 2: EXTERNAL EVENT POWER CONTROL CIRCUIT



Power consumption is dependent on the oscillator frequency of the system. The device must operate fast enough to interface with external circuitry, yet slow enough to conserve power. The designer must account for oscillator start-up time, external circuitry initialization, and code execution time when calculating device power consumption. Table 1 shows various frequency oscillators, oscillator modes and the average current consumption of each mode. A PIC16C54 was used to collect data for Table 1 and the code is shown in Example 1. A current profile for a PIC16C54 in RC oscillator mode running at 261 kHz is shown in Figure 3. Figure 4 shows a current profile for a PIC16C54 in XT mode running at 1 MHz. The current profile includes three regions: power-up, active, and sleep. The power-up region is defined as the time the PIC16/17 device is in Power-On Reset and/or Oscillator Start-up Time. The active region is the time that the PIC16/17 device is executing code and the sleep region is the time the device is in SLEEP Mode. When using a 32.768 kHz crystal in LP oscillator mode, the designer must check that the oscillator has stabilized during the Power-On Reset. Otherwise, the device may not come out of reset properly.

TABLE 1: OSCILLATOR MODES

Osc. Type	Frequency	Osc. Mode	Power-up Region Current, Time	Active Region Current, Time	Sleep Region Current, Time
Resistor / Capacitor	261 kHz	RC	51.2 µA, 17.5 ms	396 µA 12.8 ms	0.32 µA, 140 ms
Resistor / Capacitor	1.13 MHz	RC	61.4 µA, 17.5 ms	810 µA, 2.5 ms	0.3 µA, 140 ms
Crystal	32.768 kHz	LP	51.2 µA, 19 ms	23.5 µA, 93 ms	0.3 µA, 140 ms
Crystal	50 kHz	LP	61.4 µA, 16 ms	39.4 µA, 48.5 ms	0.28 µA, 140 ms
Crystal	1 MHz	XT	92 µA, 17.5 ms	443 µA, 3 ms	0.35 µA, 140 ms
Crystal	8 MHz	HS	123 µA, 18 ms	2.11 mA, 250 µs	0.3 µA, 140 ms
Resonator	455 kHz	XT	38.4 µA, 17.3 ms	421 µA, 7 ms	0.34 µA, 140 ms
Resonator	8 MHz	HS	143 µA, 18 ms	2.5 mA, 250 µs	0.29 µA, 140 ms

EXAMPLE 1: CURRENT PROFILE CODE

```
    TITLE "Current Profiling Program"
    LIST P=16C54, F=INHX8M
    INCLUDE "C:\PICMASTR\P16C5X.INC"
;***************************************************************************
;***************************************************************************
;;       This program initializes the PIC16C54, delays for 256 counts, then goes
;       to sleep. The WDT wakes up the PIC16C54.
;;*************************************************************************
;***************************************************************************
;Define General Purpose register locations
        LSB             EQU 0x10    ;delay control register
        Reset Vector
        ORG 0
START
        MOVLW           0x0B        ;WDT Prescaler of 1:8
        OPTION
        CLRF            PORTA       ;clear PORTA
        CLRF            PORTB       ;clear PORTB
        CLRW                        ;make PORTA and PORTB pins outputs
        TRIS            PORTA
        TRIS            PORTB
        CLRF            LSB
LOOP    DECFSZ          LSB,1
        GOTO            LOOP
        SLEEP                       ;go to sleep
        END
```

FIGURE 3: CURRENT PROFILE (261 kHz RC MODE)



FIGURE 4: CURRENT PROFILE (1 MHz XT MODE)



Designing a system for lower supply voltages, typically 3V, is another method to reduce I_{PD}. This type of design is best utilized in a battery powered system where current consumption is very low. A wide range of devices from op-amps and Analog-to-Digital (A/D) converters to CMOS logic products are being manufactured for low voltage operation. This gives the designer the flexibility to design a low voltage system with the same type of components that are available for a 5V design. Refer to the PIC16/17 device data sheets for I_{PD} vs. V_{DD} data.

Since any I/O pin can source or sink up to 20 mA, the PIC16/17 devices can provide power to other components. Simply connect the V_{DD} pin of an external component to an I/O pin. Currently, most of the op-amps, A/D converters, and other devices manufactured today are low-power and can be powered by this technique. This provides the ability to turn off power to sections of the system during periods of inactivity.

Temperature will effect the current consumption of the PIC16/17 devices in different ways. Typically devices will consume more current at extreme temperatures and batteries will have less available current at those same temperatures. PIC16/17 devices will exhibit higher Ipd currents at high temperatures. Refer to the PIC16/17 device data sheets for I_{PD} vs. Temperature data.

TROUBLESHOOTING IPD

The first step in troubleshooting IPD problems is to measure the IPD that the circuit is consuming. Circuits to measure IPD for all oscillator modes are shown in Figure 5 for PIC16/17 devices. The resistor Rp is used to measure the amount of current entering the VDD pin when resistor Rg is shorted. The resistor Rg is used to measure the amount of current leaving the VSS pin when resistor Rp is shorted. The value of Rp and Rg should be approximately 100Ω for all oscillator modes. The two values of current should be approximately the same when the PIC16/17 is operating at the lowest possible power. If you find that the values of IPD measured from both configurations are not equivalent or are higher than the specifications, the following suggestions should help to find the source of extra current.

FIGURE 5: CIRCUITS TO MEASURE IPD FOR PIC16/17 DEVICES



*Rs for HS and XT modes on PIC16CXX devices
XT mode on PIC17CXX devices

**PIC17C42 only

Basically, if Ip is not equal to Ig, then an I/O pin is either sourcing (Ip>Ig) current or sinking (Ip<Ig) current.

- Is the $\overline{\text{MCLR}}$ pin tied to VDD? Is the rate of rise of VDD slower than 0.05 V/ms? Does VDD start at VSS then rise? These conditions will not guarantee that the chip will come out of reset and function properly. Some of the circuits on PIC16/17 devices will start operating at lower voltage levels than other circuits. See Application Note AN522 "Power-Up Considerations" in the Microchip *Embedded Control Handbook*.
- Are all inputs being driven to VSS or VDD? If any input is not driven to either VSS or VDD, it will cause switching currents in the digital (i.e., flashing) input buffers. The exceptions are the oscillator pins and any pin configured as an analog input. During Power-on Reset or Oscillator Start-up time, pins that are floating may cause increased current consumption.
- All unused I/O pins should be configured as outputs and set high or low. This ensures that switching currents will not occur due to a floating input.
- Is the TMR0 (RTCC) pin pulled to VSS or VDD? The TMR0 pin of PIC16C5X devices should be tied to VSS or VDD for the lowest possible current consumption.
- If an analog voltage is present at a pin, is that pin configured as an analog input? If an analog voltage is present at a pin configured as a digital input, the digital input buffers devices will consume more current due to switching currents.
- Are all on-chip peripherals turned off? Any on-chip peripheral that can operate with an external clock source, such as the A/D converter or asynchronous timers, will consume extra current.
- Are you using the PORTB internal pull-up resistors? If so and if any PORTB I/O pin is driving or receiving a zero, the additional current from these resistors must be considered in the overall current consumption.
- Is the Power-Up Timer being used? This will add additional current drain during power-up.
- If the currents measured at the Rp and Rg resistors are not the same, then current is being sourced or sunk by an I/O pin. Make sure that all I/O pins that are driving external circuitry are switched to a low power state. For instance, an I/O pin that is driving an LED should be switched to a state where the LED is off.
- Is the window of a JW package device covered? Light will affect the current consumption of a JW package device with the window left uncovered.

I_{PD} Analysis Using A Random Sample

The Microchip *1994 Microchip Data Book* specifies the typical Ipd current for a PIC16C5X part at 4µA and the maximum Ipd current at 12 µA. These values are valid at a V_{DD} voltage of 3V and a temperature range of 0°C to 70°C with the Watchdog Timer enabled. A control group of fifty PIC16C54's were randomly selected with pre-production and production samples. Ipd tests were run on the group for a voltage range of 2.5V to 6.5V and for a temperature range of 0°C to 70°C. Table 2 compares the median and maximum values obtained by the Ipd tests to the typical and maximum values in the data book. The Ipd test data and the data book values are based on V_{DD} = 3.0V, Watchdog Timer Enabled, and a temperature range of 0°C to 70°C.

The values in the data book are obtained from devices in which the manufacturing process has been skewed to various extremes. This should produce devices which function close to the minimum and maximum operating ranges for each parameter shown in the data book. The typical values obtained in the data book are actually the mean value of characterization data at a temperature of 25°C. The minimum and maximum values shown in the data book are the mean value of the characterization data at the worst case temperature, plus or minus three times the standard deviation. Statistically this means that 99.5% of all devices will operate at or below the typical value and much less than the maximum value.

TABLE 2: I_{PD} COMPARISON OF CONTROL GROUP vs. DATA BOOK VALUES

Source	Typical or Median I_{PD}	Maximum
Control Group	2.349 µA	3.048 µA
1994 Microchip Data Book	4 µA	12 µA

BATTERY DESIGN

When designing a system to use batteries, the designer must consider the maximum current consumption, operating voltage range, size and weight constraints, operating temperature range, and the frequency of operation. Once the system design is finished, the designer must again ask some questions that will define what type of battery to use. What is the operating voltage range? What is the current drain rate? What are the size constraints? How long will the system be used? What type of battery costs can be tolerated? What range of temperatures will the system be operated?

It is difficult to state a rule of thumb for selecting batteries because there are many variables to consider. For example, operating voltages vary from one battery type to another. Lithium cells typically provide 3.0V while Nickel-Cadmium cells provide 1.2V. On the other hand, Lithium cells can withstand minimal discharge rates while Nickel-Cadmium can provide up to 30A of current. A designer must consider all characteristics of each battery type when making a selection. Appendix B contains a simple explanation of batteries, a characteristic table for some common battery types, and discharge curves for the common batteries.

It is very important when doing a low power design to correctly estimate the required capacity of the power source. At this point, the designer should be able to estimate the operating voltage, current drain rates and how long the system is supposed to operate. To explain how to estimate the required capacity of a system, we will use the first entry from Table 1 using an RC oscillator set at 261 kHz. Figure 3 shows the current profile for this entry. It can be seen that the profile has a period of 170.3 ms with a 17.5 ms power-up region, a 12.8 ms active region, and a 140 ms sleep region. Assuming that the system will be required to operate for six months, we can now calculate the capacity required to power this system. Example 2 will illustrate the procedure. If a system does not have a periodic current profile, then the percentages obtained in step 1 of Example 2 will have to be estimated.

EXAMPLE 2: CAPACITY CALCULATION

1. Calculate the percentage of time spent in power-up, active, and sleep regions.

   power-up
   (17.5 ms / 170.3 ms) x 100 = 10.3%

   active
   (12.8 ms / 170.3 ms) x 100 = 7.5%

   sleep
   (140 ms / 170.3 ms) x 100 = 82.2%

2. Calculate the number of hours in 6 months.

   6 months
   x (30 days / month)
   x (24 hours / day) = 4320 hours

3. Using the number of hours, percentages, and currents calculate the capacity for each period of time

   power-up
   4320 hours x 10.3% x 51.2 µA = 22.8 mAh

   active
   4320 hours x 7.5% x 396 µA = 128.3 mAh

   sleep
   4320 hours x 82.2% x 0.32 µA = 1.14 mAh

4. Sum the capacities of each period

   22.8 mAh + 128.3 mAh + 1.14 mAh = 152.2 mAh

The capacity required to operate the circuit for six months is 152.2 mAh. Example 2 does not take into consideration temperature effects or leakage currents that are associated with batteries. The load resistance of a battery is affected by temperature which in turn changes the available voltage and current; however, the self discharge rate is higher.

EXAMPLE DESIGN

A PIC16C54 with an LP oscillator of 32.768 kHz is used in this design. A Linear Technology low-power 12-bit A/D converter samples a temperature sensor. This data is transmitted via an LED at 300 baud to a receiver. The A/D converter, op-amp, and temperature sensor are powered from an I/O pin on the PIC16C54. The Watchdog Timer is enabled to periodically wake the system up from Sleep and take a sample. Figure 6 shows the schematic for the example design and Appendix A contains the source code.

This circuit has two operating modes, active and sleep. There was not a distinct power-up region in this design. In the circuit with the peripheral chips powered directly from the battery, the example design consumed 8mA of current in the active mode and 6.5 mA in SLEEP Mode. With the peripheral chips powered from an I/O pin of the PIC16C54, the example design consumed 4 mA of current in the active mode and 0.5 µA in SLEEP Mode. The advantage of using an I/O pin to provide power to peripherals can be seen in a calculation of the capacity required to operate the circuit for one month. Example 3 details the two capacity calculations.

EXAMPLE 3: CAPACITY CALCULATION FOR THE EXAMPLE DESIGN

1. Calculate the percentage of time spent in the active and SLEEP Modes.

   active - battery power
   (210 ms / 2.61 s) x 100 = 8%

   sleep - battery power
   (2.4 s / 2.61 s) x 100 = 92%

   active - I/O power
   (188 ms / 2.638 s) x 100 = 7.1%

   sleep - I/O power
   (2.45 s / 2.638 s) x 100 = 92.9%

2. Calculate the number of hours in 1 month.

   1 month
   x (30 days / month)
   x (24 hours / day)
   = 720 hours

3. Using the number of hours, percentages and currents calculate the capacity for each period of time.

   active - battery power
   720 hours x 8% x 8 mA = 461 mAh

   sleep - battery power
   720 hours x 92% x 6.5 mA = 4306 mAh

   active - I/O power
   720 hours x 7.1% x 4 mA = 205 mAh

   sleep - I/O power
   720 hours x 92.9% x 0.5 µA = 0.4 mAh

4. Sum the capacities of each period.

   battery power
   461 mAh + 4306 mAh = 4767 mAh

   I/O power
   205 mAh + 0.4 mAh = 206 mAh

The capacity required to operate this circuit for one month can be reduced by a factor of twenty just by powering the peripheral components from an I/O pin. The example design will use two Panasonic® BR2325 Lithium batteries in series to provide power to the circuit. This results in a Vbatt of 6V and a capacity of 165 mAh. Using the estimation process, the circuit should function for approximately 24 days. The actual time of operation was 24.2 days with the system running in an ambient temperature of 22°C.

FIGURE 6: EXAMPLE DESIGN SCHEMATIC



SUMMARY

This application note has described some of the methods used to lower IPD and reduce overall system current consumption. Some obvious methods such as SLEEP Mode and low voltage design were given. Techniques such as powering components from I/O pins and oscillator mode and frequency selection can also be important in reducing IPD and overall system current. Some suggestions for troubleshooting IPD problems were presented. Finally, some considerations for designing a battery powered system were offered.

Please check the Microchip BBS for the latest version of the source code. For BBS access information, see Section 6, Microchip Bulletin Board Service information, page 6-3.

APPENDIX A: EXAMPLE DESIGN CODE

```
MPASM 01.02.05 Released    LOWPWR.ASM   1-9-1995  13:2:42                 PAGE  1
Ipd/Battery Apnote Example Design
LOC  OBJECT CODE     LINE SOURCE TEXT
  VALUE

                     0001         TITLE "Ipd/Battery Apnote Example Design"
                     0002         LIST P=16C54, F=INHX8M
                     0003
                     0004         INCLUDE "C:\PICMASTR\P16C5X.INC"
                     0002 ; P16C5X.INC  Standard Header File, Version 0.1     Microchip Technology, Inc.
                     0004
                     0005
                     0006 ;****************************************************************************
                     0007 ;****************************************************************************
                     0008 ;
                     0009 ;       Filename:       lowpwr.asm
                     0010 ;       REVISION:       9 Jan 95
                     0011 ;
                     0012 ;****************************************************************************
                     0013 ;
                     0014 ;       This program initializes the PIC, takes a sample, and outputs the
                     0015 ;       value to PORTB pin 0 (the LED), and then goes to Sleep.  The
                     0016 ;       Watchdog Timer wakes the device up from Sleep.  PORTA pin 0 is used
                     0017 ;       to control power to peripherals.
                     0018 ;
                     0019 ;****************************************************************************
                     0020 ;****************************************************************************
                     0021
                     0022 ;       Define variable registers
  0010               0023         MSB             EQU     0x10
  0011               0024         LSB             EQU     0x11
  0012               0025         DELAY_CNT       EQU     0x12
  0013               0026         SHIFT           EQU     0x13
  0014               0027         COUNT           EQU     0x14
                     0028
                     0029 ;       Reset Vector
                     0030         ORG     0x1FF
  01FF 0A00          0031         GOTO    START
                     0032
                     0033 ;       Start of main code
```

```
                      ORG     0
                ;***************************************************************************
                ;       Main routine which initializes the PIC, and has main loop.
                ;***************************************************************************
0000            START
0000 0C2F               MOVLW   0x2F            ;1:128 WDT PRESCALAR
0001 0002               OPTION
0002 0C02               MOVLW   0x02            ;RA1 SET HIGH
0003 0025               MOVWF   PORTA
0004 0066               CLRF    PORTB           ;ALL PINS SET TO Vss
0005 0C08               MOVLW   0x08            ;RA3-DATA INPUT
0006 0005               TRIS    PORTA           ;RA0-POWER,RA1-CS,RA2-CLOCK OUTPUTS
0007 0040               CLRW                    ;PORTB ALL OUTPUTS, RBO-LED OUTPUT
0008 0006               TRIS    PORTB
0009 0071               CLRF    LSB             ;CLEAR A/D RESULT REGISTERS
000A 0070               CLRF    MSB

000B 0004               CLRWDT
000C 0911               CALL    SAMPLE          ;GET SAMPLE FROM A/D
000D 0004               CLRWDT
000E 0948               CALL    OUTPUT          ;OUTPUT SAMPLE TO LED AT 300 BAUD
000F 0004               CLRWDT
0010 0003               SLEEP


                ;***************************************************************************
                ;       Main routine for retrieving a sample from the A/D.
                ;***************************************************************************
0011            SAMPLE
0011 0505               BSF     PORTA,0         ;TURN POWER ON TO PERIPHERALS
0012 0943               CALL    DELAY           ;WAIT FOR PERIPHERALS TO STABILIZE
0013 0C0B               MOVLW   0x0B            ;DATA COUNTER, 12 BIT A/D
0014 0034               MOVWF   COUNT
0015 0C08               MOVLW   0x08            ;SET SHIFT REGISTER
0016 0033               MOVWF   SHIFT
0017 0000               NOP
0018 0425               BCF     PORTA,1         ;ENABLE A/D
0019 0000               NOP
001A 0545               BSF     PORTA,2         ;1ST CLOCK RISE
001B 0000               NOP
001C 0445               BCF     PORTA,2         ;1ST CLOCK FALL
001D 0000               NOP
001E 0545               BSF     PORTA,2         ;NULL BIT CLOCK RISE
001F 0000               NOP
0020 0445               BCF     PORTA,2         ;NULL BIT CLOCK FALL
0021 0000               NOP
```

```
0022 0933     LOOP    CALL    READ            ;READ DATA BIT
0023 0000             NOP
0024 0545             BSF     PORTA,2         ;BIT CLOCK RISE
0025 0000             NOP
0026 0445             BCF     PORTA,2         ;BIT CLOCK FALL
0027 0000             NOP
0028 02F4             DECFSZ  COUNT,F         ;CHECK LOOP COUNTER
0029 0A22             GOTO    LOOP
002A 0933             CALL    READ            ;READ LAST BIT
002B 0000             NOP
002C 0545             BSF     PORTA,2         ;SET CLOCK
002D 0000             NOP
002E 0525             BSF     PORTA,1         ;SET CS
002F 0000             NOP
0030 0445             BCF     PORTA,2         ;CLEAR CLOCK
0031 0405             BCF     PORTA,0         ;POWER DOWN PERIPHERALS
0032 0800             RETURN
              ;****************************************************************************
              ;       Reads a bit from PORTA, data line from the A/D.
              ;****************************************************************************
0033          READ
0033 0004             CLRWDT
0034 0774             BTFSS   COUNT,3         ;CHECK IF AT BIT 8 - 11
0035 0A3B             GOTO    RLOW            ;GOTO BITS 0 - 7
0036 0765             BTFSS   PORTA,3         ;CHECK IF DATA IS CLEAR
0037 0A3F             GOTO    REND            ;GOTO EXIT
0038 0213             MOVF    SHIFT,W         ;ADD A ONE TO MSB IN THE CORRECT
0039 01F0             ADDWF   MSB,F           ;BIT POSITION
003A 0A3F             GOTO    REND
003B 0765     RLOW    BTFSS   PORTA,3
003C 0A3F             GOTO    REND
003D 0213             MOVF    SHIFT,W         ;ADD A ONE TO LSB IN THE CORRECT
003E 01F1             ADDWF   LSB,F           ;BIT POSITION
003F 0333     REND    RRF     SHIFT,F         ;SHIFT
0040 0603             BTFSC   STATUS,C        ;IF ONE IS IN THE CARRY
0041 0333             RRF     SHIFT,F         ;SHIFT AGAIN
0042 0800             RETURN
              ;****************************************************************************
              ;       Simple delay loop for 772 clock cycles.
              ;****************************************************************************
0043          DELAY
0043 0004             CLRWDT                  ;RESET WATCHDOG TIMER
0044 0072             CLRF    DELAY_CNT
0045 02F2     DLOOPL  DECFSZ  DELAY_CNT,F
```

```
0046 0A45           GOTO    DLOOPL
0047 0800           RETURN


                 ;****************************************************************************
                 ;       Output sample to LED at 300 baud.
                 ;****************************************************************************
0048             OUTPUT
0048 0C08           MOVLW   0x08             ;SHIFT 8 MSB BITS OUT
0049 0034           MOVWF   COUNT

004A 0370        MSBOUT  RLF     MSB,F       ;SHIFT LSB INTO CARRY
004B 0703           BTFSS   STATUS,C         ;IF CARRY IS SET
004C 0A50           GOTO    MSBCLR
004D 0506           BSF     PORTB,0          ;SET PORTB,0
004E 0968           CALL    BAUD
004F 0A54           GOTO    MSBCHK           ;CHECK FOR ALL 8 BITS TO BE SENT
0050 0406        MSBCLR  BCF     PORTB,0     ;OTHERWISE CLEAR PORTB,0
0051 0000           NOP                      ;WAIT TO SET BAUD RATE 600
0052 0000           NOP
0053 0968           CALL    BAUD
0054 02F4        MSBCHK  DECFSZ  COUNT       ;CHECK FOR ALL 8 BITS TO BE SENT
0055 0A4A           GOTO    MSBOUT

0056 0C08           MOVLW   0x08             ;SHIFT 8 LSB BITS OUT
0057 0034           MOVWF   COUNT

0058 0371        LSBOUT  RLF     LSB,F       ;SHIFT LSB INTO CARRY
0059 0703           BTFSS   STATUS,C         ;IF CARRY IS SET
005A 0A5E           GOTO    LSBCLR
005B 0506           BSF     PORTB,0          ;SET PORTB,0
005C 0968           CALL    BAUD
005D 0A62           GOTO    LSBCHK           ;CHECK FOR 8 BITS TO BE SENT
005E 0406        LSBCLR  BCF     PORTB,0     ;OTHERWISE CLEAR PORTB,0
005F 0000           NOP                      ;WAIT TO SET BAUD RATE 600
0060 0000           NOP
0061 0968           CALL    BAUD
0062 02F4        LSBCHK  DECFSZ  COUNT       ;CHECK FOR 8 BITS TO BE SENT
0063 0A58           GOTO    LSBOUT
0064 0406           BCF     PORTB,0          ;CLEAR PORTB,0
0065 0071           CLRF    LSB              ;CLEAR LSB
0066 0070           CLRF    MSB              ;CLEAR MSB
0067 0800           RETURN

                 ;****************************************************************************
                 ;       Delay loop for sending data to the LED at 300 baud.
                 ;****************************************************************************
```

```
0068                BAUD
0068 0000                   NOP
0069 0000                   NOP
006A 0000                   NOP
006B 0000                   NOP
006C 0000                   NOP
006D 0000                   NOP
006E 0000                   NOP
006F 0000                   NOP
0070 0000                   NOP
0071 0000                   NOP
0072 0000                   NOP
0073 0000                   NOP
0074 0000                   NOP
0075 0800                   RETURN

                            END

MEMORY USAGE MAP ('X' = Used,  '-' = Unused)

0000 : XXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXX
0040 : XXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXX XXXXXX----------

0180 : ---------------- ---------------- ---------------- ----------------
01C0 : ---------------- ---------------- ---------------- ---------------X

All other memory blocks unused.


Errors   :    0
Warnings :    0
Messages :    0
```

APPENDIX B: BATTERY DESCRIPTIONS

Presently there are two types of batteries that are manufactured, primary and secondary. Primary batteries are those that must be thrown away once their energy has been expended. Low current drain, short duty cycles, and remote operation favor primary batteries such as Carbon Zinc and Alkaline. Secondary batteries can be recharged once they have exhausted their energy. High current drain or extended usage favors secondary batteries especially when the cost of replacement of disposable batteries is not feasible. Secondary batteries include Nickel-Cadmium and Nickel Metal Hydride.

A battery may be discharged by different means depending on the type of load. The type of load will have a significant effect on the life of the battery. The typical modes of discharge are constant resistance, constant current, and constant power. Constant resistance is when the load maintains a constant resistance throughout the discharge cycle. Constant current is the mode where the load draws the same current during discharge. Finally, constant power is defined as the current during a discharge increases as the battery voltage decreases.

The constant resistance mode results in the capacity of the battery being drained at a rapid and excessive rate, resulting in a short life. This is caused by the current during discharge following the drop in battery voltage. As a result, the levels of current and power during discharge are in excess of the minimum required.

The constant current mode has lower current and power throughout the discharge cycle than the constant resistance mode. The average current drain on the battery is lower and the discharge time to the end-voltage is longer.

The constant power discharge mode has the lowest average current drain and therefore has the longest life. During discharge, the current is lowest at the beginning of the cycle and increases as the battery voltage drops. Under this mode the battery can be discharged below its end voltage, because the current is increased as the voltage drops. The constant power mode provides the most uniform performance throughout the life of the battery and has the most efficient use of the energy in the battery.

The nominal voltage is the no-load voltage of the battery, the operating voltage is the battery voltage with a load, and the end-of-life voltage is the voltage when the battery has expended its energy. Energy Density is used to describe the amount of energy per unit of volume or mass (Wh/kg or Wh/l). Generally, energy density decreases with decreasing battery size within a particular type of battery. Most batteries are rated by an amp-hour (Ah) or milliamp-hour (mAh) rating. This rating is based on a unit of charge, not energy. A 1-amp current corresponds to the movement of 1 coulomb (C) of charge past a given point in 1 second (s). Table B-1 lists some typical characteristics of the most common types of batteries.

TABLE B-1: TYPICAL BATTERY CHARACTERISTICS

	Carbon Zinc	Alkaline	Nickel Cadmium	Lithium	Nickel Metal Hydride	Zinc Air	Silver Oxide
Cell Voltage							
Nominal	1.5	1.5	1.2	3.0	1.2	1.4	1.6
Operating	1.25-1.15	1.25-1.15	1.25-1.00	2.5-3.0	1.25-1.0	1.35-1.1	1.5
End of life	0.8	0.9	0.9	1.75	0.9	0.9	0.9
Operating Temperature	-5°C to 45°C	-20°C to 55°C	-40°C to 70°C	-30°C to 70°C	-20°C to 50°C	0°C to 45°C	-20°C to 50°C
Energy Density (Wh/kg)	70	85	30	300	55	300	100
Capacity	60mAh to 18Ah	30mAh to 45Ah	150mAh to 4Ah	35mAh to 4Ah	500mAh to 5Ah	50mAh to 520mAh	15mAh to 210mAh
Advantages		High capacity, good low temp	good low temp, good high rate discharge	good low and high temp, good high rate discharge, long shelf life	better capacity than Nicad for same size	high energy density, good shelf life	good low temp, good shelf life
Limitations	Low energy density, poor low temp, poor high rate discharge		poor low rate discharge, disposal hazards	Violent reaction to water		Cannot stop reaction once started	poor high rate discharge
Relative Cost	low	low	medium	high	high	high	high
Type	**Primary**	**Primary**	**Secondary**	**Primary**	**Secondary**	**Primary**	**Primary**

Typical discharge curves for alkaline, carbon zinc, lithium, nickel cadmium, nickel metal hydride, silver oxide, and zinc air are shown in Figure B-1 through Figure B-7. These curves are only typical representations of each battery type and are not specific to any battery manufacturer. Also the load and current drain are different for each type of battery.

FIGURE B-1: ALKALINE DISCHARGE CURVE (16 mA LOAD)



FIGURE B-2: CARBON ZINC DISCHARGE CURVE (16 mA LOAD)



FIGURE B-3: LITHIUM DISCHARGE CURVE (2.8 mA LOAD)



FIGURE B-4: NICKEL CADMIUM DISCHARGE CURVE (500 mA LOAD)



FIGURE B-5: NICKEL METAL HYDRIDE DISCHARGE CURVE (1500 mA LOAD)



FIGURE B-6: SILVER OXIDE DISCHARGE CURVE (1 mA LOAD)



FIGURE B-7: ZINC AIR DISCHARGE CURVE (1.3 mA LOAD)



NOTES:

NOTES:

NOTES:

WORLDWIDE SALES & SERVICE

AMERICAS

Corporate Office
Microchip Technology Inc.
2355 West Chandler Blvd.
Chandler, AZ 85224-6199
Tel: 602 786-7200 Fax: 602 786-7277
Technical Support: 602 786-7627
Web: http://www.mchip.com/microchip

Atlanta
Microchip Technology Inc.
500 Sugar Mill Road, Suite 200B
Atlanta, GA 30350
Tel: 770 640-0034 Fax: 770 640-0307

Boston
Microchip Technology Inc.
5 Mount Royal Avenue
Marlborough, MA 01752
Tel: 508 480-9990 Fax: 508 480-8575

Chicago
Microchip Technology Inc.
333 Pierce Road, Suite 180
Itasca, IL 60143
Tel: 708 285-0071 Fax: 708 285-0075

Dallas
Microchip Technology Inc.
14651 Dallas Parkway, Suite 816
Dallas, TX 75240-8809
Tel: 214 991-7177 Fax: 214 991-8588

Dayton
Microchip Technology Inc.
35 Rockridge Road
Englewood, OH 45322
Tel: 513 832-2543 Fax: 513 832-2841

Los Angeles
Microchip Technology Inc.
18201 Von Karman, Suite 455
Irvine, CA 92715
Tel: 714 263-1888 Fax: 714 263-1338

New York
Microchip Technology Inc.
150 Motor Parkway, Suite 416
Hauppauge, NY 11788
Tel: 516 273-5305 Fax: 516 273-5335

San Jose
Microchip Technology Inc.
2107 North First Street, Suite 590
San Jose, CA 95131
Tel: 408 436-7950 Fax: 408 436-7955

ASIA/PACIFIC

Hong Kong
Microchip Technology
Unit No. 3002-3004, Tower 1
Metroplaza
223 Hing Fong Road
Kwai Fong, N.T. Hong Kong
Tel: 852 2 401 1200 Fax: 852 2 401 3431

Korea
Microchip Technology
168-1, Youngbo Bldg. 3 Floor
Samsung-Dong, Kangnam-Ku,
Seoul, Korea
Tel: 82 2 554 7200 Fax: 82 2 558 5934

Singapore
Microchip Technology
200 Middle Road
#10-03 Prime Centre
Singapore 188980
Tel: 65 334 8870 Fax: 65 334 8850

Taiwan
Microchip Technology
10F-1C 207
Tung Hua North Road
Taipei, Taiwan, ROC
Tel: 886 2 717 7175 Fax: 886 2 545 0139

EUROPE

United Kingdom
Arizona Microchip Technology Ltd.
Unit 6, The Courtyard
Meadow Bank, Furlong Road
Bourne End, Buckinghamshire SL8 5AJ
Tel: 44 0 1628 851077 Fax: 44 0 1628 850259

France
Arizona Microchip Technology SARL
2 Rue du Buisson aux Fraises
91300 Massy - France
Tel: 33 1 69 53 63 20 Fax: 33 1 69 30 90 79

Germany
Arizona Microchip Technology GmbH
Gustav-Heinemann-Ring 125
D-81739 Muenchen, Germany
Tel: 49 89 627 144 0 Fax: 49 89 627 144 44

Italy
Arizona Microchip Technology SRL
Centro Direzionale Colleoni
Palazzo Pegaso Ingresso No. 2
Via Paracelso 23, 20041
Agrate Brianza (MI) Italy
Tel: 39 039 689 9939 Fax: 39 039 689 9883

JAPAN

Microchip Technology Intl. Inc.
Benex S-1 6F
3-18-20, Shin Yokohama
Kohoku-Ku, Yokohama
Kanagawa 222 Japan
Tel: 81 45 471 6166 Fax: 81 45 471 6122

9/22/95

MICROCHIP